In[1]:=
```
<< IntervalTools`
```

Example: $g(a, T)$ from Counting Zeros of Dirichlet L-Functions

In “Counting Zeros of Dirichlet L-Functions”, we define a function $g(a, T) := \frac{2}{\pi}\,\mathrm{Im}\left[\mathrm{LogGamma}\left[\frac{1}{4} + \frac{a}{2} + I * \frac{T}{2}\right]\right] - \frac{T}{\pi}\,\mathrm{Log}\left[\frac{T}{2E}\right] - \frac{2a-1}{4}$. Using Stirling’s Approximation, we get upper and lower bounds on $g(a, T)$, and need to prove the simplifying $\frac{a-2}{50T} \le g(a, T) \le \frac{2-a}{50T}$, for $a \in \{0, 1\}$ and $T \ge 5/7$. That’s 4 inequalities, if you are counting: upper and lower bounds for $a = 0$ and for $a = 1$.

In[2]:=
```
Plot[{(0 - 2)/50, T * (2/π Im[LogGamma[1/4 + 0/2 + I * T/2]] - T/π Log[T/(2 E)] - (2 × 0 - 1)/4), (2 - 0)/50},
 {T, 5 / 7, 4}, PlotLegends → {(2 - a)/(50 T), "T g(a,T)", (a - 2)/(50 T)}, PlotLabel → "a=0"]
```

Out[2]=



In[3]:=
$$\texttt{Plot}\left[\left\{\frac{1-2}{50},\ T*\left(\frac{2}{\pi}\,\texttt{Im}\left[\texttt{LogGamma}\left[\frac{1}{4}+\frac{1}{2}+I*\frac{T}{2}\right]\right]-\frac{T}{\pi}\,\texttt{Log}\left[\frac{T}{2\,E}\right]-\frac{2\times 1-1}{4}\right),\ \frac{2-1}{50}\right\},\right.$$
$$\left.\{T,\ 5/7,\ 4\},\ \texttt{PlotLegends}\to\left\{\frac{2-a}{50\,T},\ \texttt{"T g(a,T)"},\ \frac{a-2}{50\,T}\right\},\ \texttt{PlotLabel}\to\texttt{"a=1"}\right]$$

Out[3]=



The pieces

The upper bound on $T*g(a,T)$ is

$$\frac{T}{240\left(\left(\frac{9}{4}+\frac{a}{2}\right)^2+\frac{T^2}{4}\right)^{3/2}}+\frac{T}{180\,\pi\left(\left(\frac{9}{4}+\frac{a}{2}\right)^2+\frac{T^2}{4}\right)^{3/2}}-$$
$$\frac{T^2}{12\,\pi\left(\left(\frac{9}{4}+\frac{a}{2}\right)^2+\frac{T^2}{4}\right)}+\frac{2\,T\,\text{ArcTan}\left[\frac{2\left(\frac{1}{4}+\frac{a}{2}\right)}{T}\right]}{\pi}+\frac{2\,T\,\text{ArcTan}\left[\frac{2\left(\frac{5}{4}+\frac{a}{2}\right)}{T}\right]}{\pi}-$$
$$\frac{7\,T\,\text{ArcTan}\left[\frac{2\left(\frac{9}{4}+\frac{a}{2}\right)}{T}\right]}{2\,\pi}-\frac{a\,T\,\text{ArcTan}\left[\frac{2\left(\frac{9}{4}+\frac{a}{2}\right)}{T}\right]}{\pi}+\frac{T^2\,\text{Log}\left[1+\frac{4\left(\frac{9}{4}+\frac{a}{2}\right)^2}{T^2}\right]}{2\,\pi}$$

Which we presciently express as

In[4]:=
$$\texttt{g11[a\_, T\_]} = \frac{T/2}{\left(\left(\frac{9}{4}+\frac{a}{2}\right)^2+\frac{T^2}{4}\right)^{3/2}};$$

$$\texttt{g12[a\_, T\_]} = \frac{T^2/4}{\left(\left(\frac{9}{4}+\frac{a}{2}\right)^2+\frac{T^2}{4}\right)};$$

$$\texttt{g13[c\_, T\_]} = T\,\texttt{ArcTan}\left[\frac{c}{T}\right];$$

$$\texttt{g14[a\_, T\_]} = T^2\,\texttt{Log}\left[1+\frac{4\left(\frac{9}{4}+\frac{a}{2}\right)^2}{T^2}\right];$$

$$\texttt{gUpper[a\_, T\_]} := \left(\frac{1}{120}+\frac{1}{90\,\pi}\right)\texttt{g11[a, T]}-\frac{1}{3\,\pi}\,\texttt{g12[a, T]}+$$
$$\frac{1}{\pi}\left(2\,\texttt{g13}\left[\frac{1}{2}+a,\ T\right]+2\,\texttt{g13}\left[\frac{5}{2}+a,\ T\right]-\left(\frac{7}{2}+a\right)\texttt{g13}\left[\frac{9}{2}+a,\ T\right]\right)+\frac{1}{2\,\pi}\,\texttt{g14[a, T]};$$

In[9]:=
```
Manipulate[Plot[
  {gUpper[a, T], T * (2/π Im[LogGamma[1/4 + a/2 + I * T/2]] - T/π Log[T/(2 E)] - (2 a - 1)/4)}, {T, 5 / 7, 4},
  PlotRange → All, PlotLegends → {"upper bound", "T g(a,T)"}], {a, {0, 1}}]
```

Out[9]=



We now proceed to find sharper interval enclosures for g11, g12, g13, g14. Both g11 and g12 are essentially polynomials in one variable, and so are easy to handle. The arctangent presents some awkwardness in g13, but the logarithm in g14 is easy.

g11

In[10]:=
```
(*g11[a,T] has only one positive  critical value *)
Solve[D[g11[a, T], T] == 0]
```

Out[10]= $\left\{\left\{T \to -\frac{9+2a}{2\sqrt{2}}\right\}, \left\{T \to \frac{9+2a}{2\sqrt{2}}\right\}\right\}$

In[11]:=
```
g11[a_, Tint_Interval] :=
  PiecewiseMonotoneEnclosure[(T / 2)/(((9/4 + a/2)^2 + T^2/4)^(3/2)), T, {(9 + 2 a)/(2 √2)}, Tint];
```

g12

In[12]:=
```
(*g12[a,T] has no positive critical values *)
Solve[D[g12[a, T], T] == 0]
```

Out[12]= $\left\{\left\{a \to -\frac{9}{2}\right\}, \{T \to 0\}\right\}$

In[13]:=
```
g12[a_, Tint_Interval] := MonotoneEnclosure[(T^2/4)/(((9/4 + a/2)^2 + T^2/4)), T, Tint];
```

g13 (the hard one)

In[14]:=
```
(*g13 is somewhat trickier, but it is also strickly increasing. *)
(* We care about 1/2≤c≤11/2 and T≥5/7 *)
D[g13[c, T], T] /. c → x T // Simplify
```

Out[14]= $-\frac{x}{1+x^2} + \text{ArcTan}[x]$

As $x = \frac{c}{T}$, we should consider $0 < x \leq \frac{11/2}{5/7} = \frac{77}{10}$. We need to show on this domain that

$$\text{ArcTan}[x] - \frac{x}{1+x^2} > 0.$$

In[15]:=
```
Plot[ArcTan[x] - x/(1 + x^2), {x, 0, 77/10}]
```

Out[15]=


We can restrict our attention to $0 < x < \infty$.
By Alirezaei's Arctangent Bound,

$$\text{ArcTan}[x] - \frac{x}{1+x^2} \geq \frac{x}{\frac{4}{\pi^2}+\text{Sqrt}\left[\left(1-\frac{4}{\pi^2}\right)^2+\frac{4}{\pi^2}x^2\right]} - \frac{x}{1+x^2} = \frac{x\left(1-\frac{4}{\pi^2}+x^2-\sqrt{\left(1-\frac{4}{\pi^2}\right)^2+\frac{4}{\pi^2}x^2}\right)}{(1+x^2)\left(\frac{4}{\pi^2}+\sqrt{\left(1-\frac{4}{\pi^2}\right)^2+\frac{4}{\pi^2}x^2}\right)}.$$

Clearly x, $1+x^2$, and $\frac{4}{\pi^2} + \sqrt{\left(1-\frac{4}{\pi^2}\right)^2 + \frac{4}{\pi^2}x^2}$ are positive for positive x. The remaining term is positive as $x \to \infty$, is continuous, and is never 0 (a polynomial root problem):

In[16]:=
```
Solve[1 - 4/π^2 + x^2 - Sqrt[(1 - 4/π^2)^2 + (4 x^2)/π^2] == 0]
```

Out[16]= {{x → 0}}

In[17]:=
```
g13[c_, Tint_Interval] := MonotoneEnclosure[T ArcTan[c/T], T, Tint];
```

g14

In[18]:=
```
g14[a_, T_] = T^2 Log[1 + 4 (9/4 + a/2)^2 / T^2];
```

In[19]:=
```
Simplify[D[g14[a, T], T]]
```

Out[19]= $2\,T\left(-\frac{(9+2\,a)^2}{81+36\,a+4\,a^2+4\,T^2}+\text{Log}\left[1+\frac{(9+2\,a)^2}{4\,T^2}\right]\right)$

A picture showing the standard bounds on logarithms.

In[20]:=
```
Plot[{x/(1 + x), Log[1 + x], x}, {x, -1/2, 1/2}]
```

Out[20]=


Recall that $\frac{x}{1+x} \leq \text{Log}[1+x]$. For positive T, the derivative $D[\text{g14}[a, T], T]$ has the same sign as

$$-\frac{(9+2\,a)^2}{81+36\,a+4\,a^2+4\,T^2}+\text{Log}\left[1+\frac{(9+2\,a)^2}{4\,T^2}\right] \geq -\frac{(9+2\,a)^2}{81+36\,a+4\,a^2+4\,T^2}+\frac{\frac{(9+2\,a)^2}{4\,T^2}}{1+\frac{(9+2\,a)^2}{4\,T^2}}=0.$$

In[21]:=
```
Simplify[-(9 + 2 a)^2/(81 + 36 a + 4 a^2 + 4 T^2) + ((9+2 a)^2/(4 T^2))/(1 + (9+2 a)^2/(4 T^2))]
```

Out[21]= 0

In[22]:=
```
Limit[g14[a, T], T → Infinity]
```

Out[22]= $\frac{1}{4}(9+2\,a)^2$

In[23]:=
```
(*we thus  have the following interval version of g14*)

g14[a_, Tint_Interval] := MonotoneEnclosure[T^2 Log[1 + 4 (9/4 + a/2)^2 / T^2], T, Tint];
```

The proof for the upper bound

In[24]:=
```
{fails, uncertains, proved} = ProveNonNegative[(2 - 0)/50 - gUpper[0, #] &,
   (2 - 0)/50 - gUpper[0, #] &, {Interval[{5 / 7, Infinity}]}, MaxDepth → 50];
fails
uncertains
Length[proved]
Last[proved]
```

Out[25]= { }

Out[26]= { }

Out[27]= 361

Out[28]= {Interval[{64, ∞}]}

So the proof for $a = 0$ works, splitting the interval $\left[\frac{5}{7}, \infty\right]$ into 361 pieces, with the infinite one being $[64, \infty]$.

In[29]:=
```
{fails, uncertains, proved} = ProveNonNegative[(2 - 1)/50 - gUpper[1, #] &,
   (2 - 1)/50 - gUpper[1, #] &, {Interval[{5 / 7, Infinity}]}, MaxDepth → 50];
{fails, uncertains}
{Length[proved], Last[proved]}
```

Out[30]= {{ }, { }}

Out[31]= {1277, {Interval[{128, ∞}]}}

We have $T * g(a, T) \leq \frac{2-a}{50}$.

The lower bound

The expression from Stirling's Approximation that we need to show is at least $\frac{a-2}{50}$ is

In[32]:=
```
gLower[a_, T_] := (-1/120 + -1/(90 π)) g11[a, T] - 1/(3 π) g12[a, T] +
   1/π (2 g13[1/2 + a, T] + 2 g13[5/2 + a, T] - (7/2 + a) g13[9/2 + a, T]) + 1/(2 π) g14[a, T];
```

In[33]:=
```
Manipulate[Plot[
  {gLower[a, T], T * (2/π Im[LogGamma[1/4 + a/2 + I * T/2]] - T/π Log[T/(2 E)] - (2 a - 1)/4)}, {T, 5 / 7, 4},
  PlotRange → All, PlotLegends → {"lower bound", "T g(a,T)"}], {a, {0, 1}}]
```

Out[33]=



The proof for the lower bound

In[34]:=
```
{fails, uncertains, proved} = ProveNonNegative[(gLower[0, #] + (2 - 0)/50) &,
    (gLower[0, #] + (2 - 0)/50) &, {Interval[{5 / 7, Infinity}]}, MaxDepth → 15];
{fails, uncertains}
{Length[proved], Last[proved]}
```

Out[35]= `{{}, {}}`

Out[36]= `{109, {Interval[{32, ∞}]}}`

In[37]:=
```
{fails, uncertains, proved} = ProveNonNegative[((2 - 1)/50 + gLower[1, #]) &,
    ((2 - 1)/50 + gLower[1, #]) &, {Interval[{5 / 7, Infinity}]}, MaxDepth → 15];
{fails, uncertains}
{Length[proved], Last[proved]}
```

Out[38]= `{{}, {}}`

Out[39]= `{329, {Interval[{64, ∞}]}}`